EVALUATION OF A PROPOSED MODIFIED LOG-GAMMA CONFIDENCE BOUND METHOD FOR FLEET MISSILE SYSTEM RELIABILITY

Peter Allen Craig

NAVAL POSTGRADUATE SCHOOL
Monterey, California

THESIS

EVALUATION OF A PROPOSED MODIFIED
LOG-GAMMA CONFIDENCE BOUND METHOD
FOR FLEET MISSILE SYSTEM RELIABILITY

by

Peter Allen Craig

September 1979

Thesis Advisor: M. Woods

Approved for public release; distribution unlimited.

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) Evaluation of a Proposed Modified Log-Gamma Confidence Bound Method for Fleet Missile System Reliability		5. TYPE OF REPORT & PERIOD COVERED Master's Thesis; September 1979
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Peter Allen Craig		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Naval Postgraduate School Monterey, California 93940		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS
11. CONTROLLING OFFICE NAME AND ADDRESS Naval Postgraduate School Monterey, California 93940		12. REPORT DATE September 1979
		13. NUMBER OF PAGES 46
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Fleet Missile System Reliability
Log-Gamma Confidence Bound

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

A statistical method is evaluated to determine its accuracy for estimating lower confidence bounds on system reliability of a mixture of missile configurations using component data.

#20 - ABSTRACT - CONTINUED

Monte Carlo simulations are used to establish the accuracy of these bounds.

EVALUATION OF A PROPOSED MODIFIED LOG-GAMMA CONFIDENCE BOUND METHOD FOR FLEET MISSILE SYSTEM RELIABILITY

by

Peter Allen Craig
Lieutenant, U. S. Navy
B.S., Marquette University, 1973

Submitted in partial fulfillment of the
requirements for the degree of

MASTER OF SCIENCE IN OPERATIONS RESEARCH

from the

NAVAL POSTGRADUATE SCHOOL
September 1979

ABSTRACT

A statistical method is evaluated to determine its accuracy for estimating lower confidence bounds on system reliability of a mixture of missile configurations using component data. Monte Carlo simulations are used to establish the accuracy of these bounds.

TABLE OF CONTENTS

LIST OF TABLES

I. INTRODUCTION

A statistical method has been proposed which obtains a lower confidence bound on system reliability. It is a modified log-gamma procedure developed to measure fleet missile system reliability. Monte Carlo simulations were performed to evaluate its accuracy as an estimate for system reliability. Five hundred simulations were run for each of twelve cases examined at 80% and 90% confidence levels. The results of these simulations are included in this paper. Additional simulations were performed with minor modifications to the proposed log-gamma method. These changes are documented and the results are included. A comparison was made between the two versions on their accuracy for estimating the lower confidence bound on system reliability.

The reliability equations were applied to a hypothetical fleet missile system configuration and analyzed for changes in test sample sizes, component reliabilities and weighting factors. The proposed procedure was determined to be significantly inaccurate for small and large amounts of accumulated test data on missile components. It also has the distracting defect that larger lower confidence bounds are obtained from data sets with one failure than those obtained from data sets with zero failures.

II. MODIFIED LOG-GAMMA METHOD

The log-gamma method, in its more general form, can apply to nonseries as well as to fleet-mixture populations. The underlying theory is contained in [Ref. 1]. Examples of cases where it is suspect have been included in the following chapter. The procedure below describes the proposed modified log-gamma method as it is applied to a series system.

Assume that in a series system there are k components each with a sample size n_i, where $i = 1, 2, \ldots, k$. Let the number of failures be f_i for $i = 1, 2, \ldots, k$. Consider first the case when there is at least one failure in the system. Thus $\Sigma f_i > 0$.

Let

$$\hat{R}_i = 1 - \frac{f_i}{n_i} \tag{2.1}$$

be the point estimates of the i-th component reliability. Then the equation

$$\hat{R} = \prod_{i=1}^{k} \hat{R}_i \tag{2.2}$$

is the point estimate of the system reliability. Define

$$\overline{R} = R^{1/k} \tag{2.3}$$

and

$$\hat{V} = (1 - \bar{R}) \sum_{i=1}^{k} \frac{1}{n_i} \qquad (2.4)$$

$\hat{V}$ is used as an estimate of the variance of $-\ln \hat{R}$. It is assumed that the distribution of $-\ln \hat{R}$ can be approximated by a gamma distribution as follows

$$f(z) = \frac{z^{L-1} e^{Lz/\ln R}}{(\frac{-\ln R}{L}) \Gamma(L)} , \quad z \geq 0 \qquad (2.5)$$

where $z = -\ln \hat{R}$ and L and $(\frac{-\ln R}{L})$ are parameters.

Let

$$L^* = \frac{(-\ln \hat{R})^2}{\hat{V}} \qquad (2.6)$$

and

$$\hat{L} = L^* + 2.25 \qquad (2.7)$$

L^* is the method-of-moments estimate of the shape parameter. A constant term 2.25 is added to L^*, the shape parameter estimate in the proposed modified log-gamma procedure. The lower $(1-\alpha)$ confidence bound, $\underline{R}(1-\alpha)$ is given by solving the equation

$$\underline{R}(1-\alpha) = \hat{R}^{(2\hat{L}/\chi^2_{2\hat{L},\alpha})} \qquad (2.8)$$

where $\chi^2_{2\hat{L},\alpha}$ is the lower α-quantity of the chi-square distribution with $2\hat{L}$ degrees of freedom. Interpolation is required if $2\hat{L}$ is noninteger.

If there are zero failures in the system ($\sum f_i = 0$), let

$$N^* = \frac{k}{\sum_{i=1}^{k} \frac{1}{n_i}} \tag{2.9}$$

where N* is defined to be the effective sample size. Then the lower $1-\alpha$ confidence bound $\underline{R}(1-\alpha)$ is computed according to a binomial confidence bound based on zero failures out of N* trials (i.e., $\underline{R}(1-\alpha) = \sqrt[N^*]{\alpha}$). If N* is noninteger then linear interpolation is recommended in the proposed procedure but it is not necessary because the same formula could be used for N* an integer.

The modified log-gamma method has been described here for both zero failures and one or more failures in series. The more general form of this method was applied to an actual missile system configuration to determine the lower confidence bounds. The program used to evaluate its accuracy has been included in Appendix B. The complete listing and definitions of the variables used in the program are listed in Appendix A. A description of the more generalized method is described as it was applied to the specific missile system simulated.

In the fleet missile system examined there were different groups of missiles with different configurations. The population was therefore not homogeneous and weights were

assigned to the different groups. There were 14 components in the system modeled and eight mixture weights for the subgroups. The input data consisted of f_i (the number of failures in the i-th component), n_i (the sample size for the i-th component), M_i (the exponent of each component) and C_j (the weights applied to each subgroup). Point estimates for this system were defined as follows

$$\hat{R}_i = 1 - \frac{f_i}{n_i} \tag{2.10}$$

$$\hat{p}_R = \prod_{i=1}^{5} R_i^{M_i} \tag{2.11}$$

$$\hat{p}_N = \prod_{i=6}^{10} R_i^{M_i} \tag{2.12}$$

with the subgroup reliability point estimates being

$$\begin{aligned}
\hat{R}^{(1)} &= \hat{p}_R \hat{R}_{11} \hat{R}_{12} \\
\hat{R}^{(2)} &= \hat{p}_N \hat{R}_{11} \hat{R}_{12} \\
\hat{R}^{(3)} &= \hat{p}_R \hat{R}_{13} \hat{R}_{14} \\
\hat{R}^{(4)} &= \hat{p}_N \hat{R}_{13} \hat{R}_{14} \\
\hat{R}^{(5)} &= \hat{p}_R \hat{R}_{13} \hat{R}_{12} \\
\hat{R}^{(6)} &= \hat{p}_N \hat{R}_{13} \hat{R}_{12}
\end{aligned} \tag{2.13}$$

$$\hat{R}^{(7)} = \hat{p}_R \hat{R}_{11} \hat{R}_{14}$$

$$\hat{R}^{(8)} = \hat{p}_N \hat{R}_{11} \hat{R}_{14}$$

and

$$\hat{R} = \sum_{j=1}^{8} c_j \hat{R}^{(j)} \tag{2.14}$$

The variance of $-\ln \hat{R}$ is then estimated by $\hat{V}$ given by equation (2.15)

$$\hat{V} = \frac{1}{\hat{R}^2} \sum_{i=1}^{8} \sum_{j=1}^{8} c_i c_j \hat{R}^{(i)} \hat{R}^{(j)} S_{ij} \tag{2.15}$$

where S_{ij} estimates the $\mathrm{cov}(z^{(i)}, z^{(j)})$ and where $z^{(i)} = -\ln \hat{R}^{(i)}$. The estimates S_{ij} are found by solving the following equations.

$$z_i = -\ln \hat{R}_i \tag{2.16}$$

$$\bar{R} = \exp\left(-\sum_{i=1}^{14} M_i z_i \Big/ \sum_{i=1}^{14} M_i\right) \tag{2.17}$$

and

$$V_R = (1 - \bar{R}) \sum_{i=1}^{5} \frac{M_i^2}{n_i} \tag{2.18}$$

$$V_N = (1 - \bar{R}) \sum_{i=6}^{10} \frac{M_i^2}{n_i} \tag{2.19}$$

$$V_i = (1 - \bar{R})/n_i \quad , \quad i = 11,\ldots,14 \tag{2.20}$$

Then the S_{ij}'s are solved by the equations listed in the program in Appendix B and repeated below.

$$
\begin{aligned}
S(1,1) &= V_R + V_{11} + V_{12} & S(2,6) &= V_N + V_{12} \\
S(2,2) &= V_N + V_{11} + V_{12} & S(2,7) &= V_{11} \\
S(3,3) &= V_R + V_{13} + V_{14} & S(2,8) &= V_N + V_{11} \\
S(4,4) &= V_N + V_{13} + V_{14} & S(3,4) &= V_{13} + V_{14} \\
S(5,5) &= V_R + V_{13} + V_{12} & S(3,5) &= V_R + V_{13} \\
S(6,6) &= V_N + V_{13} + V_{12} & S(3,6) &= V_{13} \\
S(7,7) &= V_R + V_{11} + V_{14} & S(3,7) &= V_R + V_{14} \\
S(8,8) &= V_N + V_{11} + V_{14} & S(3,8) &= V_{14} \\
S(1,2) &= V_{11} + V_{12} & S(4,5) &= V_{13} \\
S(1,3) &= V_R & S(4,6) &= V_N + V_{13} \\
S(1,4) &= 0 & S(4,7) &= V_{14} \\
S(1,5) &= V_R + V_{12} & S(4,8) &= V_N + V_{14} \\
S(1,6) &= V_{12} & S(5,6) &= V_{13} + V_{12} \\
S(1,7) &= V_R + V_{11} & S(5,7) &= V_R \\
S(1,8) &= V_{11} & S(5,8) &= 0 \\
S(2,3) &= 0 & S(6,7) &= 0 \\
S(2,4) &= V_N & S(6,8) &= V_N \\
S(2,5) &= V_{12} & S(7,8) &= V_{11} + V_{14}
\end{aligned}
\tag{2.21}
$$

Finally,

$$\hat{L} = \frac{(-\ln \hat{R})^2}{\hat{V}} + 2.25 \tag{2.22}$$

and DF, the degrees of freedom, is equal to

$$DF = 2\hat{L} \tag{2.23}$$

Thus

$$\underline{R}(1-\alpha) = \hat{R}^{(DF/\chi^2_{DF,\alpha})} \tag{2.24}$$

III. EVALUATION PROCEDURE

The equation for system reliability is

$$R_s = \sum_{j=1}^{L} w_j \prod_{i=1}^{k} p_i^{M_i} \qquad (3.1)$$

where

L = number of subsystems

w_j = the weighting factor of the j-th subsystem

k = the number of components

p_i = the reliability of the i-th component

M_i = the exponent of the i-th component

The computer program modeled a system that had 8 subsystems and 14 components. System reliability (RS) was determined for each case and a lower confidence bound for $\alpha = .1$ and $\alpha = .2$ was computed. Random numbers were drawn from a shuffled random number generator [Ref. 3]. Inverse chi-square values were determined using the international mathematical and statistical library (IMSL) routine called MDCHI. All computations were done in single precision arithmetic, coded in FORTRAN, using an IBM 360 computer.

A. ZERO FAILURE VS ONE FAILURE CASE

An examination of two cases revealed a shortcoming and a motivation for evaluating the modified log-gamma procedure. These two examples are considered below.

Example 1.

Let k, the number of components in the system, be 14 and let R_i, the component reliabilities, all equal .99. The sample sizes (mission trials) and failures for each component are listed in Table I. The lower 90% confidence limit on system reliability is desired.

Table I

	Component													
	1	2	3	4	5	6	7	8	9	10	11	12	13	14
M_i: # mission trials	10	10	10	10	500	10	10	10	10	500	10	10	10	10
f_i: # failures	0	0	0	0	0	0	0	0	0	0	0	0	0	0

When the $\Sigma f_i = 0$ the modified log-gamma procedure defines N*, the effective sample size, as

$$N^* = \frac{k}{\sum_{i=1}^{k} \frac{1}{n_i}} \tag{3.2}$$

For the data given in the table above N* is equal to 11.628. For this procedure the lower 1-α confidence bound $\underline{R}(1-\alpha)$ is computed according to a binomial confidence bound based on zero failures out of N* trials. The value obtained for $\underline{R}(1-\alpha)$ was .820.

Example 2.

Let sample sizes and f_i (failures for each component) be given in Table II. Again the lower 90% confidence limit on system reliability is desired.

Table II

	Component													
	1	2	3	4	5	6	7	8	9	10	11	12	13	14
M_i: # mission trials	10	10	10	10	500	10	10	10	10	500	10	10	10	10
f_i: # failures	0	0	0	0	1	0	0	0	0	0	0	0	0	0

When $\Sigma f_i \neq 0$ the modified log-gamma method solves for $\underline{R}(1-\alpha)$ (the lower confidence bound) by the following procedure.

Let

$$\hat{R}_i = 1 - \frac{f_i}{n_i} ,$$

$$\hat{R}_5 = .998 , \qquad (3.3)$$

$$\hat{R}_i = 1 \quad \text{for} \quad i \neq 5$$

be the point estimate of the i-th component reliability.

Then

$$\hat{R} = \prod_{i=1}^{k} R_i = .998 \qquad (3.4)$$

is the point estimate for system reliability. Define

$$\bar{R} = \hat{R}^{1/k} = .99986 \qquad (3.5)$$

$$\hat{V} = (1-\bar{R}) \sum_{i=1}^{k} \frac{1}{n_i} = .000172 \qquad (3.6)$$

where $\hat{V}$ estimates the variance of $-\ln \hat{R}$.

Let

$$L^* = \frac{(-\ln \hat{R})^2}{\hat{V}} = .02328 \qquad (3.7)$$

and define

$$\hat{L} = L^* + 2.25 = 2.27328 \qquad (3.8)$$

where 2.25 is the correction term and L^* is the method-of-moments estimate of the shape parameter. Then the lower $1-\alpha$ confidence bound, $\underline{R}(1-\alpha)$ is computed by solving

$$\underline{R}(1-\alpha) = \hat{R}^{(2L/\chi^2_{2\hat{L},\alpha})} \qquad (3.9)$$

where $\chi^2_{2\hat{L},\alpha}$ is the lower α quantity of the chi-square distribution with $2\hat{L}$ degrees of freedom. In example 2 $\underline{R}(1-\alpha)$ is equal to .993

These two examples have shown the shortcoming of this method. The lower confidence bound for one failures is higher than the lower confidence bound for zero failures.

B. SIMULATION RESULTS

The lower confidence bound values obtained for the twelve cases studied have been listed in Table III. RS is the system reliability, ACV is the actual confidence value computed by the modified log-gamma method and $\underline{R}(1-\alpha)*500$ is the percentile value of the 500 ordered $\underline{R}(1-\alpha)$ estimates for $\alpha = .1$ and $\alpha = .2$. N(I), RI(I) and W(I) are the respective sample sizes, reliabilities and weights assigned to each case.

For example, in case number 3 the number of components k, is equal to 14 with the sample sizes equal to 50. for $i \neq 5$ or 10 and 250 for $i = 5$ or 10. The reliabilities of each component is .99 and the 8 weights are all equal to .125. System reliability, R_S, was computed to be .816 and for $\alpha = .1$ the 450-th value in the ordered 500 LCL estimates was .895. The R_S value of .816 was the 35th of the 500 ordered LCL estimates yielding an actual confidence level of 7.8%. Likewise for $\alpha = .2$ the 400-th value in the ordered 500 LCL estimates was .898. The R_S value of .816 was the 13-th of the 500 ordered LCL estimates yielding an actual confidence level of 2.8%. In only one case (case 8) did the actual confidence value approach that of the system reliability as a lower bound.

An examination of the MLG (modified log-gamma) procedure questioned the inclusion of the correction term 2.25. Additional simulations were run on the same twelve cases when this correction term was removed and the degrees of freedom bounded

below by 1.0. The results obtained from this modification, while an improvement, were still far from providing accurate lower bounds on the system. The values determined from these runsa're listed in Table IV. ACV values of 100% indicate that the system reliability was greater than all 500 estimates.

It would appear that in order to generate more estimates less than RS the exponent, $2L/\chi^2_{2\hat{L},\alpha}$, needs to take on larger values. Adding a constant term such as 2.25 yields more values for $\underline{R}(1-\alpha)$ that are greater than RS. Indeed, Tables III and IV did show this to be the case. As the exponent becomes larger (the chi-squared value smaller) the confidence level decreases. The estimate for $\hat{L}$ used in generating the values listed in Table IV seem more accurate when used in the modified log-gamma procedure.

This modification still left much room for improvement. A closer reivew of the MLG method pointed to the estimate of the shape parameter as a possible cause of the extreme results. Since $Z = -\ln \hat{R}$ its distribution was approximated by a two-parameter gamma distribution. Then

$$f(z) = \frac{z^{L-1}\, e^{Lz/\ln R}}{(-\frac{\ln R}{L})^{L}\, \Gamma(L)} \quad , \quad z \geq 0 \qquad (3.10)$$

where L and $(-\frac{\ln R}{L})$ are the parameters. Then

$$E(z) = L \cdot \frac{(-\ln R)}{L} = -\ln R \qquad (3.11)$$

and

$$\mathrm{Var}(z) = L\left(\frac{-\ln R}{L}\right)^2 = \frac{\ln^2 R}{L} \tag{3.12}$$

Note:

$$L = \frac{\ln^2 R}{\mathrm{Var}(z)} = \frac{[E(z)]^2}{\mathrm{Var}(z)} \tag{3.13}$$

The proposed estimator $\hat{L}$ for L is

$$\hat{L}_1 = \frac{z^2}{\widehat{\mathrm{Var}}(z)} \tag{3.14}$$

Since $L = \dfrac{[E(z)]^2}{\mathrm{Var}(z)}$ it would appear that $\hat{L} = \dfrac{[\widehat{E(z)}]^2}{\widehat{\mathrm{Var}}(z)}$ would be a better estimator for L. Since

$$[E(z)]^2 = E(z^2) - \mathrm{Var}(z)$$

we have

$$\hat{L} = \frac{\widehat{E(z^2)} - \widehat{\mathrm{Var}}(z)}{\widehat{\mathrm{Var}}(z)} \tag{3.15}$$

and since z^2 is unbiased for $E(z^2)$ we get

$$\hat{L} = \frac{z^2 - \widehat{\mathrm{Var}}(z)}{\widehat{\mathrm{Var}}(z)} = \frac{z^2}{\widehat{\mathrm{Var}}(z)} - 1 \tag{3.16}$$

Note that this is a departure from $\hat{L}_1$ in the proposed method. Thus the shape parameter L can be estimated by Eq. 3.16 above. This estimate is different from the original version of the MLG method.

Substituting this new value for L and bounding the degrees of freedom by 1.0, so as not to obtain a negative value, the results show a little more improvement. The results obtained from this second modification are listed in Table V.

Accuracy of $\underline{R}(1-\alpha)$ as a $100(1-\alpha)\%$ Lower Confidence Limit for RS

(Correction Term Equal to 2.25)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF $(1-\alpha)$*500	ACV	STANDARD DEVIATION OF R $(1-\alpha)$
1	14	N(I)=10, I=1,2,...,14 EXCEPT N(5)=50, N(10)=50	.816	.1 .2	.872 .889	66.4% 29.4%	.068 .062
		RI(I)=.99, I=1,2,...,14					
		W(J)=.125, J=1,2,...,8					
2	14	N(I)=20, I=1, 2,...,14 EXCEPT N(5)=100 AND N(10)=100	.816	.1 .2	.900 .903	40.2% 20.8%	.052 .048
		RI(I)=.99, I=1,2,...,14					
		W(J)=.125, J=1,2,...,8					
3	14	N(I)=50, I=1, 2,...,14 EXCEPT N(5)=250 AND N(10)=250	.816	.1 .2	.895 .898	7.8% 2.8%	.029 .028
		RI(I)=.99, I=1,2,...,14					
		W(J)=.125, J=1,2,...,8					
4	14	N(I)=10, I=1, 2,...,14 EXCEPT N(5)=80, N(10)=80	.816	.1 .2	.903 .889	43.4% 27.2%	.067 .061
		RI(I)=.99, I=1,2,...,14					
		W(J)=.125, J=1,2,...,8					

TABLE III (Continued)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
5	14	N(I)=20, I=1,2,...,14 EXCEPT N(5)=160 AND N(10)=160 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.898 .900	30.6% 12.2%	.048 .044
6	14	N(I)=50, I=1,2,...,14 EXCEPT N(5)=400 AND N(10)=400 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.895 .898	5.6% 1.0%	.026 .025
7	14	N(I): 5 4 7 RI(I):.998 .99 .995 N(I): 9 20 5 4 RI(I):.98 .997 .998 .99 N(I): 7 9 20 5 RI(I):.995 .998 .998 .99 N(I): 4 7 9 RI(I):.993 .98 .99 W(J):.0625 .125 .125 .125 .0625 .250 .125 .125	.883	.1 .2	.923 .878	83.0% 82.8%	.090 .070

TABLE III (Continued)

CASE NO.	K	N(I), RI(I) AND W(J)	RS	α ALPHA	R OF $(1-\alpha)$*500	ACV	STANDARD DEVIATION OF R $(1-\alpha)$
8	14	N(I): 20 16 28 36 RI(I): .998 .99 .995 .98	.883	.1 .2	.931 .938	44.0% 26.4%	.041 .036
		N(I): 80 20 16 28 RI(I): .997 .998 .99 .995					
		N(I): 36 80 20 16 RI(I): .998 .998 .99 .993					
		N(I): 28 36 RI(I): .98 .99					
		W(J): .0625 .125 .125 .125 .0625 .250 .125 .125					
9	14	N(I): 50 40 70 90 RI(I): .998 .99 .995 .98	.883	.1 .2	.931 .937	20.6% 5.4%	.023 .021
		N(I): 200 50 40 70 RI(I): .997 .998 .99 .995					
		N(I): 90 200 50 40 RI(I): .998 .998 .99 .993					
		N(I): 70 90 RI(I): .98 .99					
		W(J): .0625 .125 .125 .125 .0625 .250 .125 .125					

TABLE III (Continued)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	F CF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
10	14	N(I): 15 50 20 30 RI(I):.995 .99 .957 .98 N(I): 100 5 20 10 RI(I):.998 .99 .993 .99 N(I): 20 100 15 30 RI(I):.997 .993 .97 .99 N(I): 8 7 RI(I):.98 .97 W(J)=.125, J=1,2,...,8	.821	.1 .2	.886 .897	38.4% 16.8%	.049 .043
11	14	N(I): 15 50 20 30 RI(I):.985 .98 .987 .97 N(I): 100 5 20 10 RI(I):.988 .98 .983 .98 N(I): 20 100 15 30 RI(I):.987 .983 .96 .98 N(I): 8 7 RI(I):.97 .96 W(J)=.125, J=1,2,...,8	.669	.1 .2	.787 .801	23.4% 7.4%	.057 .055
12	14	N(I): 15 50 20 RI(I):.9975 .995 .9985 N(I): 30 100 5 20 RI(I):.99 .999 .995 .9965 N(I): 10 20 100 RI(I):.995 .985 .9965 N(I): 15 30 8 7 RI(I):.985 .995 .99 .985 W(J)=.125, J=1,2,...,8	.907	.1 .2	.936 .939	61.6% 43.2%	.038 .031

Accuracy of $\underline{R}(1-\alpha)$ as a $100(1-\alpha)\%$ Lower Confidence Limit for RS

(Correction term equal to 0.0, DF $\geq$ 1.0)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	ALPHA	R OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
1	14	N(I)=10, I=1,2,...,14 EXCEPT N(5)=5C, N(10)=50 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.816 .859	100.% 69.4%	.148 .074
2	14	N(I)=20, I=1,2,...,14 EXCEPT N(5)=100 AND N(10)=100 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.854 .871	69.8% 31.6%	.C85 .045
3	14	N(I)=50, I=1,2,...,14 EXCEPT N(5)=250 AND N(10)=250 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.886 .894	10.0% 3.2%	.028 .C27
4	14	N(I)=10, I=1,2,...,14 EXCEPT N(5)=80, N(10)=80 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.818 .857	86.4% 59.4%	.157 .C74

TABLE IV (Continued)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
5	14	N(I)=20, I=1,2,...,14 EXCEPT N(5)=160 AND N(10)=160 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.863 .882	54.8% 19.6%	.051 .041
6	14	N(I)=50, I=1,2,...,14 EXCEPT N(5)=400 AND N(10)=400 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.888 .894	7.4% 1.2%	.025 .024
7	14	N(I): 5 4 7 RI(I):.998 .99 .995 N(I): 9 20 5 4 RI(I):.98 .997 .998 .99 N(I): 7 9 20 5 RI(I):.995 .998 .998 .99 N(I): 4 7 9 RI(I):.993 .98 .99 W(J): .0625 .125 .125 .125 .0625 .250 .125 .125	.883	.1 .2	.695 .775	100.% 100.%	.241 .130

TABLE IV (Continued)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
8	14	N(I): 20 16 28 36 RI(I): .998 .99 .995 .98	.883	.1 .2	.887 .906	88.0% 61.8%	.146 .057
		N(I): 80 20 16 28 RI(I): .997 .998 .99 .995					
		N(I): 36 80 20 16 RI(I): .998 .998 .99 .993					
		N(I): 28 36 RI(I): .98 .99					
		W(J): .0625 .125 .125 .125 .0625 .250 .125 .125					
9	14	N(I): 50 40 70 90 RI(I): .998 .99 .995 .98	.883	.1 .2	.917 .928	43.0% 8.2%	.022 .020
		N(I): 200 50 40 70 RI(I): .997 .998 .99 .995					
		N(I): 90 200 50 40 RI(I): .998 .998 .99 .993					
		N(I): 70 90 RI(I): .98 .99					
		W(J): .0625 .125 .125 .125 .0625 .250 .125 .125					

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF $(1-\alpha)$*500	ACV	STANDARD DEVIATION OF R $(1-\alpha)$
10	14	N(I): 15 50 20 30 RI(I):.995 .99 .997 .98	.821	.1 .2	.860 .874	74.2% 50.4%	.148 .075
		N(I): 100 5 20 10 RI(I):.998 .99 .993 .99					
		N(I): 20 100 15 30 RI(I):.997 .993 .97 .99					
		N(I): 8 7 RI(I):.98 .97					
		W(J)=.125, J=1,2,...,8					
11	14	N(I): 15 50 20 30 RI(I):.985 .98 .987 .97	.669	.1 .2	.759 .778	40.6% 11.0%	.062 .053
		N(I): 100 5 20 10 RI(I):.988 .98 .983 .98					
		N(I): 20 100 15 30 RI(I):.987 .983 .96 .98					
		N(I): 8 7 RI(I):.97 .96					
		W(J)=.125, J=1,2,...,8					
12	14	N(I): 15 50 20 RI(I):.9975 .995 .9985	.907	.1 .2	.888 .906	95.8% 80.6%	.245 .094
		N(I): 30 100 5 20 RI(I):.99 .999 .995 .9965					
		N(I): 10 20 100 RI(I):.995 .985 .9965					
		N(I): 15 30 8 7 RI(I):.985 .995 .99 .985					
		W(J)=.125, J=1,2,...,8					

TABLE V

Accuracy OF $\underline{R}(1-\alpha)$ as a $100(1-\alpha)\%$ Lower Confidence Limit for RS

(Correction term equal to -1.0; DF $\geq$ 1.0)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
1	14	N(I)=10, I=1,2,...,14 EXCEPT N(5)=50, N(10)=50 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.816 .752	100.% 88.8%	.242 .159
2	14	N(I)=20, I=1,2,...,14 EXCEPT N(5)=100 AND N(10)=100 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.833 .833	86.6% 53.8%	.180 .078
3	14	N(I)=50, I=1,2,...,14 EXCEPT N(5)=250 AND N(10)=250 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.880 .891	11.8% 3.2%	.027 .026
4	14	N(I)=10, I=1,2,...,14 EXCEPT N(5)=80, N(10)=80 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.700 .799	94.6% 86.4%	.245 .147

TABLE V (Continued)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	F OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
5	14	N(I)=20, I=1,2,...,14 EXCEPT N(5)=160 AND N(10)=160 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.848 .867	74.4% 33.6%	.137 .069
6	14	N(I)=50, I=1,2,...,14 EXCEPT N(5)=400 AND N(10)=400 RI(I)=.99, I=1,2,...,14 W(J)=.125, J=1,2,...,8	.816	.1 .2	.884 .892	7.8% 1.2%	.024 .024
7	14	N(I): 5 4 7 RI(I): .998 .99 .995 N(I): 9 20 5 4 RI(I): .98 .997 .998 .99 N(I): 7 9 20 5 RI(I): .995 .998 .998 .99 N(I): 4 7 9 RI(I): .993 .98 .99 W(J): .0625 .125 .125 .125 .0625 .250 .125 .125	.883	.1 .2	.695 .775	100.% 100.%	.303 .225

TABLE V (Continued)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
8	14	N(I): 20 16 28 36 RI(I):.998 .99 .995 .98	.883	.1 .2	.866 .869	97.8% 84.0%	.295 .134
		N(I): 80 20 16 28 RI(I):.997 .998 .99 .995					
		N(I): 36 80 20 16 RI(I):.998 .998 .99 .993					
		N(I): 28 36 RI(I):.98 .99					
		W(J):.0625 .125 .125 .125 .0625 .250 .125 .125					
9	14	N(I): 50 40 70 90 RI(I):.998 .99 .995 .98	.883	.1 .2	.908 .919	62.0% 13.6%	.052 .026
		N(I): 200 50 40 70 RI(I):.997 .998 .99 .995					
		N(I): 90 200 50 40 RI(I):.998 .998 .99 .993					
		N(I): 70 90 RI(I):.98 .99					
		W(J):.0625 .125 .125 .125 .0625 .250 .125 .125					

TABLE [illegible] (Continued)

CASE NO.	K	N(I),RI(I) AND W(J)	RS	α ALPHA	R OF (1-α)*500	ACV	STANDARD DEVIATION OF R (1-α)
10	14	N(I): 15 50 20 30	.821	.1	.854	80.0%	.294
		RI(I): .995 .99 .997 .98		.2	.860	65.8%	.179
		N(I): 100 5 20 10					
		RI(I): .998 .99 .993 .99					
		N(I): 20 100 15 30					
		RI(I): .997 .993 .97 .99					
		N(I): 8 7					
		RI(I): .98 .97					
		W(J)=.125, J=1,2,...,8					
11	14	N(I): 15 50 20 30	.669	.1	.744	56.6%	.096
		RI(I): .985 .98 .987 .97		.2	.764	17.8%	.070
		N(I): 100 5 20 10					
		RI(I): .988 .98 .983 .98					
		N(I): 20 100 15 30					
		RI(I): .987 .983 .96 .98					
		N(I): 8 7					
		RI(I): .97 .96					
		W(J)=.125, J=1,2,...,8					
12	14	N(I): 15 50 20	.907	.1	.854	100.%	.334
		RI(I): .9975 .995 .9985		.2	.896	95.8%	.163
		N(I): 30 100 5 20					
		RI(I): .99 .999 .995 .9965					
		N(I): 10 20 100					
		RI(I): .995 .985 .9965					
		N(I): 15 30 8 7					
		RI(I): .985 .995 .99 .985					
		W(J)=.125, J=1,2,...,8					

IV. CONCLUSIONS

Additional simulations on many more cases would be required to determine the particular conditions under which this modified log-gamma method is reasonably accurate. For the cases examined here the proposed procedure remains suspect in estimating lower confidence bounds on system reliability.

APPENDIX A

AA	CORRECTION TERM EQUAL TO 2.25 IN THE MODIFIED LOG-GAMMA METHOD
AB	VARIABLE THAT STORES THE DIFFERENCE BETWEEN RS (SYSTEM RELIABILITY) AND RR(400)-- THE 80TH PERCENTILE POINT WHEN ALPHA=0.2
ABS	ABSOLUTE VALUE
AC	VARIABLE THAT STORES THE DIFFERENCE BETWEEN RS (SYSTEM RELIABILITY) AND R(450)-- THE 90TH PERCENTILE POINT WHEN ALPHA=0.1
ALOG	NATURAL LOGARITHM SUBROUTINE
ALPHA	VARIABLE ASSIGNED A VALUE OF 0.1
ALPHAA	VARIABLE ASSIGNED A VALUE OF 0.2
AM	ARRAY THAT STORES THE EXPONENTS---M SUB I
BLHAT	VARIABLE THAT STORES THE L HAT VALUE
CA	ACTUAL CONFIDENCE LEVEL FOR ALPHA=0.1
CALL	FORTRAN CODE FOR ACCESSING SUBROUTINES
CB	ACTUAL CONFIDENCE LEVEL FOR ALPHA=0.2
CONTINUE	FORTRAN CODE TO CLOSE EACH DO LOOP
DA	DUMMY VARIABLE USED TO DETERMINE THE ACTUAL CONFIDENCE LEVEL
DDF	DEGREES OF FREEDOM
DIMENSION	FORTRAN CODE REQUIRED FOR DIMENSIONING ARRAYS
DO	FORTRAN CODE USED TO BEGIN LOOPS
DUM	DUMMY VARIABLE
EA	DUMMY VARIABLE USED TO DETERMINE ACTUAL CONFIDENCE VALUE
EFFN	EFFECTIVE SAMPLE SIZE
END	FORTRAN CODE REQUIRED TO END PROGRAM
EXP	EXPONENTIAL SUBROUTINE
FA	DUMMY VARIABLE USED TO DETERMINE THE ACTUAL CONFIDENCE VALUE
FLOAT	FORTRAN CODE USED TO CHANGE INTEGERS TO DECIMAL VALUES
FORMAT	FORTRAN STATEMENT
GA	DUMMY VARA
GA	DUMMY VARIABLE USED TO DETERMINE THE ACTUAL CONFIDENCE VALUE
GO	FORTRAN CODE USED IN THE --GO TO--STATEMENT
HISTG	SUBROUTINE WHICH GENERATES A HISTOGRAM OF THE DATA AND SUMMARY STATISTICS

I	INDEX VARIABLE
IER	ERROR VARIABLE IN SUBROUTINE MDCHI
IF	VARIABLE USED TO STORE THE NUMBER OF FAILURES (ALSO PART OF THE FORTRAN --IF-- STATEMENT)
II	INDEX VARIABLE
ISEED	VARIABLE THAT STORES THE INITIAL VALUE FOR CALLING RANDOM NUMBERS
J	INDEX VARIABLE
JF	VARIABLE THAT STORES THE NUMBER OF FAILURES PER COMPONENT
JJ	INDEX VARIABLE
JM	INDEX VARIABLE
K	VARIABLE THAT STORES THE NUMBER OF COMPONENTS (ALSO USED AS AN INDEXING VARIABLE)
KJ	INDEX VARIABLE
L	VARIABLE THAT STORES THE NUMBER OF SUBSYSTEMS (ALSO USED AS AN INDEXING VARIABLE)
MC	COUNTER VARIABLE
MDCHI	INVERSE CHI SQUARE SUBROUTINE
MM	INDEX VARIABLE
N	ARRAY THAT STORES THE K SAMPLE SIZES
NC	COUNTER VARIABLE
NCASE	VARIABLE THAT STORES THE NUMBER OF CASES
NN	INDEX VARIABLE
OVFLOW	SUBROUTINE REQUIRED FOR RANDOM NUMBER GENERATION
P	ARRAY THAT STORES THE UNIFORM RANDOM NUMBERS
PD	VARIABLE THAT STORES THE ALPHA VALUE OF .1
PN	VARIABLE THAT STORES A POINT ESTIMATE
PR	VARIABLE THAT STORES A POINT ESTIMATE
R	ARRAY THAT STORES THE LOWER CONFIDENCE BOUND VALUE WHEN ALPHA=0.1
RB	VARIABLE THAT STORES THE R(450) VALUE
RBAR	VARIABLE THAT STORES R BAR
RBN	NUMBER OF REENTRY BODIES PER MISSILE
RDEN	DUMMY VARIABLE USED TO COMPUTE RBAR
READ	FORTRAN STATEMENT
REFFN	VARIABLE THAT STORES THE INVERSE OF THE EFFECTIVE SAMPLE SIZE
RHAT	VARIABLE THAT STORES THE SUM OF THE WEIGHTED SUBGROUP RELIABILITY ESTIMATES

RI	ARRAY THAT STORES THE INPUTED RELIABILITY VALUES
RIHAT	ARRAY THAT STORES THE COMPUTED RELIABILITY VALUES
RMEAN	VARIABLE THAT STORES THE MEAN OF THE R ARRAY
RNUM	DUMMY VARIABLE USED TO COMPUTE RBAR
RR	ARRAY THAT STORES THE LOWER CONFIDENCE BOUNDS WHEN ALPHA=0.2
RRB	VARIABLE THAT STORES THE RR(400) VALUE
RRMEAN	VARIABLE THAT STORES THE MEAN OF THE RR ARRAY
RRVAR	VARIABLE THAT STORES THE VARIANCE OF THE RR ARRAY
RS	VARIABLE THAT STORES THE TOTAL SYSTEM RELIABILITY
RUHAT	ARRAY THAT STORES THE SUBGROUP RELIABILITY ESTIMATES
RVAR	VARIABLE THAT STORES THE VARIANCE OF THE R ARRAY
S	ARRAY THAT STORES THE VAR/COV MATRIX
SDR	VARIABLE THAT STORES THE STANDARD DEVIATION OF THE R ARRAY
SDRR	VARIABLE THAT STORES THE STANDARD DEVIATION OF THE RR ARRAY
SQRT	SUBROUTINE THAT SOLVES SQUARE ROOTS
SRAND	SUBROUTINE THAT IS THE SHUFFLED RANDOM NUMBER GENERATOR
STOP	FORTRAN REQUIRED CODE
SUM	DUMMY VARIABLE USED THROUGHOUT THE PROGRAM
TO	PART OF THE FORTRAN --GO TO-- STATEMENT
V	ARRAY THAT STORES 4 VARIANCE ESTIMATES FOR COMPONENTS 11 THROUGH 14
VHAT	VARIABLE THAT STORES THE VARIANCE ESTIMATE FOR -LN(RHAT)
VN	VARIABLE THAT STORES THE VARIANCE ESTIMATE FOR -LN(PN)
VR	VARIABLE THAT STORES THE VARIANCE ESTIMATE FOR -LN(PR)
VX	DUMMY VARIABLE USED IN THE MDCHI SUBROUTINE
VY	DUMMY VARIABLE USED IN THE MDCHI SUBROUTINE
W	ARRAY THAT STORES THE WEIGHTED VALUES OF EACH SUBSYSTEM
WRITE	FORTRAN STATEMENT
Z	DUMMY VARIABLE USED TO DETERMINE RBAR

APPENDIX B

```
      DIMENSION R(500),RI(50),W(50),N(50),RIHAT(50),AM(50),
     1RUHAT(50),V(4),S(8,8),RR(500),P(500)
      CALL CVFLOW
      NCASE = 0
      ISEED = 134869
C
C
C     READING IN THE INPUT PARAMETERS K,L,RBN,A,ALPHA AND
C     ALPHAA
C
C
      READ (5,330) K,L,RBN,AA,ALPHA,ALPHAA
C
C
C     READING IN  N(I)--- THE NUMBER OF RANDOM NUMBERS PER
C     COMPONENT
C
C
   20 READ (5,340) (N(I),I=1,K)
C
C
C     READING IN THE COMPONENT/FUNCTION RELIABILITIES
C
C
      READ (5,350) (RI(I),I=1,K)
      NCASE = NCASE+1
      IF (ISEED.GT.134869) GO TO 30
C
C
C     READING IN THE EXPONENTS M SUB I
C
C
      READ (5,360) (AM(I),I=1,K)
C
C
C     READING IN THE WEIGHTS FOR EACH SUBSYSTEM/GROUP
C
C
   30 READ (5,370) (W(I),I=1,L)
C
C
C     STARTING THE MAIN LOOP FOR 500 SIMULATIONS
C
C
C
      DO 190 I=1,500
C
C     LOOPING FOR EACH COMPONENT
C     AND DRAWING THE RANDOM NUMBERS
C
      IF = 0
C
      DO 50 J=1,K
      JJ = N(J)
      CALL SRAND (ISEED,P,JJ)
      JF = 0
C
      DO 40 JM=1,JJ
      IF (P(JM).GT.RI(J)) JF=JF+1
   40 CONTINUE
C
      IF = IF+JF
C
C     VARIABLE "IF" COUNTS THE FAILURES
C
      RIHAT(J) = 1.-(FLOAT(JF)/FLOAT(JJ))
   50 CONTINUE
C
```

```
      IF (IF.EQ.0) GO TO 170
C
C     COMPUTING THE POINT ESTIMATES
C
      PN = 1.
      PR = 1.
C
C
      DO 60 J=1,5
      PR = (RIHAT(J)**AM(J))*PR
      PN = (RIHAT(J+5)**AM(J+5))*PN
   60 CONTINUE
C
C
C
C     COMPUTING THE SUBGROUP RELIABILITY ESTIMATES
C
C
      RUHAT(1) = PR*RI(11)*RI(12)
      RUHAT(2) = PN*RI(11)*RI(12)
      RUHAT(3) = PR*RI(13)*RI(14)
      RUHAT(4) = PN*RI(13)*RI(14)
      RUHAT(5) = PR*RI(13)*RI(12)
      RUHAT(6) = PN*RI(13)*RI(12)
      RUHAT(7) = PR*RI(11)*RI(14)
      RUHAT(8) = PN*RI(11)*RI(14)
C
C     COMPUTING RHAT AND CALLING IT BY THE SAME NAME---RHAT
C
      RHAT = 0.
C
      DO 70 J=1,L
      RHAT = (W(J)*RUHAT(J))+RHAT
   70 CONTINUE
C
C
C
C     ESTIMATING THE VARIANCE OF -LN(RHAT)-------VHAT
C
C
C     STEP 1:  DETERMINING RBAR
C
      RNUM = 0.
      RDEN = 0.
C
C
      DO 80 J=1,K
      RNUM = (-ALOG(RIHAT(J))*AM(J))+RNUM
      RDEN = AM(J)+RDEN
   80 CONTINUE
C
      Z = (-RNUM)/RDEN
      IF (Z.LT.0) GO TO 90
      RBAR = EXP(Z)
      GO TO 100
   90 RBAR = 1./EXP(ABS(Z))
C
C     STEP 2:  DETERMINING THE VARIANCE ESTIMATES
C
  100 VR = 0.
      VN = 0.
C
      DO 110 J=1,5
      VR = (AM(J)**2/FLOAT(N(J)))+VR
      VN = (AM(J+5)**2/FLOAT(N(J+5)))+VN
  110 CONTINUE
C
      VR = (1.-RBAR)*VR
      VN = (1.-RBAR)*VN
C
      DO 120 J=1,4
      V(J) = (1.-RBAR)/FLOAT(N(J+10))
```

```
  120 CONTINUE
C
C
C
C     STEP 3:  FINAL SOLUTIONS FOR COVARIANCE ESTIMATES
C
C
      S(1,1)  =  VR+V(1)+V(2)
      S(2,2)  =  VN+V(1)+V(2)
      S(3,3)  =  VR+V(3)+V(4)
      S(4,4)  =  VN+V(3)+V(4)
      S(5,5)  =  VR+V(3)+V(2)
      S(6,6)  =  VN+V(3)+V(2)
      S(7,7)  =  VR+V(1)+V(4)
      S(8,8)  =  VN+V(1)+V(4)
      S(1,2)  =  V(1)+V(2)
      S(1,3)  =  VR
      S(1,4)  =  0.
      S(1,5)  =  VR+V(2)
      S(1,6)  =  V(2)
      S(1,7)  =  VR+V(1)
      S(1,8)  =  V(1)
      S(2,3)  =  0.
      S(2,4)  =  VN
      S(2,5)  =  V(2)
      S(2,6)  =  VN+V(2)
      S(2,7)  =  V(1)
      S(2,8)  =  VN+V(1)
      S(3,4)  =  V(3)+V(4)
      S(3,5)  =  VR+V(3)
      S(3,6)  =  V(3)
      S(3,7)  =  VR+V(4)
      S(3,8)  =  V(4)
      S(4,5)  =  V(3)
      S(4,6)  =  VN+V(3)
      S(4,7)  =  V(4)
      S(4,8)  =  VN+V(4)
      S(5,6)  =  V(3)+V(2)
      S(5,7)  =  VR
      S(5,8)  =  0.
      S(6,7)  =  0.
      S(6,8)  =  VN
      S(7,8)  =  V(1)+V(4)
C
C     FILLING IN THE REST OF THE VAR/COVAR MATRIX
C
C
      DO 140 MM=1,L
C
      DO 130 NN=1,L
      S(NN,MM)  =  S(MM,NN)
  130 CONTINUE
C
  140 CONTINUE
C
C
C
C     SOLVING THE OVERALL EQUATION FOR VHAT
C
C
      VHAT  = 0.
C
      DO 160 J=1,L
C
      DO 150 KJ=1,L
      VHAT = W(J)*W(KJ)*RUHAT(J)*RUHAT(KJ)*S(J,KJ)+VHAT
  150 CONTINUE
C
  160 CONTINUE
C
      VHAT  =  VHAT/(RHAT**2)
C
```

```
C
C     COMPUTING LHAT
C
C
      BLHAT=((-ALOG(RHAT))**2/VHAT)-1.0
C
C
C     COMPUTING THE DEGREES OF FREEDOM-DDF AND SOLVING FOR R
C     OF (1-ALPHA) WHEN THE SUM OF THE FAILURES DOES NOT
C     EQUAL ZERO
C
C
      DDF = 2.*BLHAT
      IF (DDF.LT.1.0) DDF=1.0
      PD = .1
      CALL MDCHI (PD,DDF,VX,IER)
      PD = .2
      CALL MDCHI (PD,DDF,VY,IER)
      R(I) = RHAT**((DDF)/VX)
      RR(I) = RHAT**((DDF)/VY)
      GO TO 190
C
C
C     COMPUTING RELIABILITY ESTIMATES WHEN THE SUM OF THE
C     FAILURES IS GREATER THAN ZERO
C
C
  170 SUM = 0.
      DO 180 II=1,K
      SUM = SUM+(1./FLOAT(N(II)))
  180 CONTINUE
      EFFN = FLOAT(K)/SUM
      REFFN = 1./EFFN
      R(I) = ALPHA**REFFN
      RR(I) = ALPHAA**REFFN
      GO TO 190
  190 CONTINUE
C
C
C
      CALL HISTG (R,500,0)
      CALL HISTG (RR,500,0)
C
C
C     COMPUTING THE TOTAL SYSTEM RELIABILITY
C
C
      DUM = 0.
C
      DO 210 J=1,L
      SUM = 1.
C
      DO 200 I=1,K
      SUM = SUM*(RI(I)**AM(I))
  200 CONTINUE
C
      DUM = (W(J)*SUM)+DUM
  210 CONTINUE
C
      RS = DUM
C
C
      AC = RS-R(450)
      AB = RS-RR(400)
      RB = R(450)
      RRB = RR(400)
C
C
C     COMPUTING THE SAMPLE VARIANCE FOR ALPHA=.1--SDR
C     AND ALPHA=.2---SDRR
C
C
```

```
      SUM = 0.
      DUM = 0.
C
      DO 220 I=1,500
      SUM = R(I)+SUM
      DUM = RR(I)+DUM
  220 CONTINUE
C
      RMEAN = SUM/500.
      RRMEAN = DUM/500.
      SUM = 0.
      DUM = 0.
C
      DO 230 I=1,500
      SUM = (R(I)-RMEAN)**2+SUM
      DUM = (RR(I)-RRMEAN)**2+DUM
  230 CONTINUE
C
      RVAR = SUM/499.
      RRVAR = DUM/499.
      SDR = SQRT(RVAR)
      SDRR = SQRT(RRVAR)
C
C
C     DETERMINING THE ACTUAL CONFIDENCE VALUE
C
C
      MC = 0
C
      DO 250 I=1,500
      IF (RS.GT.R(I)) GO TO 240
      DA = R(I)-RS
      EA = RS-R(MC)
      IF (DA.LT.EA) GO TO 260
      CA = FLOAT(MC)/500.
      GO TO 270
  240 MC = MC+1
  250 CONTINUE
C
  260 CA = FLOAT(I)/500.
  270 CA = CA*100.
C
C
C
      NC = 0
C
      DO 290 I=1,500
      IF (RS.GT.RR(I)) GO TO 280
      FA = RR(I)-RS
      GA = RS-RR(NC)
      IF (FA.LT.GA) GO TO 300
      CB = FLOAT(NC)/500.
      GO TO 310
  280 NC = NC+1
  290 CONTINUE
C
  300 CB = FLOAT(I)/500.
  310 CB = CB*100.
C
C
C     PRINTING THE FINAL RESULTS FOR EACH CASE
C
C
      WRITE (6,380) NCASE
      WRITE (6,390)
      WRITE (6,400) RS,RB,AC,SDR,CA
      WRITE (6,410)
      WRITE (6,420) RS,RRB,AB,SDRR,CB
      ISEED = ISEED+27
      IF (NCASE.EQ.8) GO TO 320
      GO TO 20
  320 STOP
```

```
C
C
  330 FORMAT (I2,I2,F5.2,F4.2,2F3.1)
  340 FORMAT (14I3)
  350 FORMAT (10F6.4/4F6.4)
  360 FORMAT (14F4.2)
  370 FORMAT (8F6.4)
  380 FORMAT ('1',T62,'CASE  ',I2)
  390 FORMAT ('0',///'0',T30,'ALPHA= .1')
  400 FORMAT ('0',T35,'RS= ',F10.8,T55,'R(450)= ',F10.8,T75,
     1'RS-R(450)= ',F10.8//'0',T35,'STANDARD DEVIATION = ',
     1F10.8//'0',T35,'ACTUAL CONFIDENCE VALUE= ',F6.2,' %')
  410 FORMAT ('0',////'0',T30,'ALPHA= .2')
  420 FORMAT ('0',T35,'RS= ',F10.8,T55,'RR(400)= ',F10.8,T75
     1,'RS-RR(400)= ',F10.8//'0',T35,'STANDARD DEVIATICN= ',
     1F10.8//'0',T35,'ACTUAL CONFIDENCE VALUE= ',F6.2,' %')
      END
```

BIBLIOGRAPHY

1. Chow, T., Schiller, L. and Tomsky, J., "System Reliability Estimation from Several Data Sets," Proceedings 1976 Annual Reliability and Maintainability Symposium, pp. 18-24.

2. Degroot, Morris H., Probability and Statistics, Addison-Wesley, 1975.

3. Naval Postgraduate School Research Report NPS55LW73061A, Naval Postgraduate School Random Number Generator Package, by G.P. Learmonth and P.A.W. Lewis, June 1973.

INITIAL DISTRIBUTION LIST

		No. Copies
1.	Library, Code 0142 Naval Postgraduate School Monterey, California 93940	2
2.	W.M. Woods, Code 011 Dean of Educational Development Naval Postgraduate School Monterey, California 93940	1
3.	Lt. Peter Allen Craig, USN 11 Webster Street Pringle, Pennsylvania 18709	1
4.	Defense Documentation Center Cameron Station Alexandria, Virginia 22314	2
5.	Department Chairman, Code 55 Department of Operations Research Naval Postgraduate School Monterey, California 93940	1